# $Id$
# Maintainer:  Ionut Biru <ibiru@archlinux.org>
# Contributor: Daniel Balieiro <daniel@balieiro.com>

pkgname=telepathy-salut
pkgver=0.8.1
pkgrel=4
pkgdesc="A link-local XMPP connection manager for Telepathy"
arch=('i686' 'x86_64')
url="http://telepathy.freedesktop.org"
groups=('telepathy')
license=('LGPL')
depends=('telepathy-glib' 'avahi' 'libsoup' 'util-linux')
makedepends=('libxslt' 'python2' 'xmldiff')
install=telepathy-salut.install
source=(http://telepathy.freedesktop.org/releases/$pkgname/$pkgname-$pkgver.tar.gz)
md5sums=('7516e6f6fa56a61054413a03642b938d')

build() {
    cd "$pkgname-$pkgver"
    ./configure --prefix=/usr \
    	--libexecdir=/usr/lib/telepathy \
        --disable-plugins --disable-Werror \
        --with-tls=gnutls \
        --enable-avahi-tests=no \
        --enable-static=no
    make
}

package() {
    cd "$pkgname-$pkgver"
    make DESTDIR="$pkgdir" install
}
